Pensieve header: Non-Commutative Gaussian Elimination - Day 2.

Today. Some older challenges and a new one, then commutative Gaussian elimination, then non-commutative Gaussian elimination (NCGE), then EIWL-10-12, then even less likely, **Patterns**.

Topics (in no particular order). Whatever you may suggest; whatever comes to my mind; ~~the Fibonacci numbers~~; ~~the Catalan numbers~~; ~~the Jones polynomial~~; ~~a more efficient Jones algorithm~~; ~~a riddle on spheres~~; Khovanov homology; Γ-calculus; the Hopf fibration; Hilbert's 13th problem; non-commutative Gaussian elimination; free Lie algebras; the Baker-Campbell-Hausdorff formula; wacky numbers; an order 4 torus; the Schwarz Lantern; knot colourings; the Temperley-Lieb pairing; the dodecahedral link; ~~sound experiments~~; barycentric subdivisions; ~~some Peano curves~~; braid closures and Vogel's algorithm; ~~the insolubility of the quintic~~; phase portraits; the Mandelbrot set; shadows of the Cantor aerogel; quilt plots; some image transformations; De Bruijn graphs; the Riemann series theorem; finite type invariants and the Willerton fish; ~~the Towers of Hanoi~~; ~~Hochschild homology of (some) coalgebras~~; convolutions and image improvements; the 8-5-3 milk jug problem; a cow problem.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Classes\\17-1750-ShamelessMathematica\\"];
```

```
Play[Sin[440 × 2 Pi t] + Sin[441 × 2 Pi t], {t, 0, 10}]
```



An Image Manipulation Challenge

The image at http://drorbn.net/bbs/show?shot=17-1750-171016-111042.jpg is pathetic. Can you improve it? Whatever you do, should also work well with all other images at http://drorbn.net/bbs/show.php?prefix=17-1750.

Solution by Charlene Chu:

```
img1 =
```



```
MakePrintable[img1]
```



```
NotebookOpen[
  Directory[] <> "\\StudentProjects\\chu_charlene_171105_animagemanipulationchallenge.nb"];
```

A Graphics Challenge

The torus $S^1 \times S^1$ has an order 4 symmetry. Can you draw it in such a manner that it will be manifest?

Solution by Tristan Milne:



```
NotebookOpen[Directory[] <> "\\StudentProjects\\Milne_Tristan_171106_TorusEmbeddings2.nb"];
```

The Cow Problem (Leonard)

A farmer has 19 cows, and she wishes to give them to her daughters to that the first will get 1/2, the second 1/4, and the third 1/5. This is obviously impossible. A wise woman hears of the problem and suggests: “I’ll add on one of my cows, so you’ll have 20. The first daughter will get 10, the second 5, the third 4, I’ll take back the remaining cow, and everyone is happy!”.

Problem. Are there any other quadruples like (19, 2, 4, 5), for which the same trick will work? What are all of them?

Hint. It is sometimes better to analyze and generalize, first.

Solution by Charlene Chu:

n	x	y	z	$n+1=\frac{n+1}{x}+\frac{n+1}{y}+\frac{n+1}{z}+1$?
3	4	4	4	True
5	2	6	6	True
5	3	3	6	True
7	2	4	8	True
9	2	5	5	True
11	2	3	12	True
11	2	4	6	True
11	3	3	4	True
17	2	3	9	True
19	2	4	5	True
23	2	3	8	True
41	2	3	7	True

```
NotebookOpen[Directory[] <> "\\StudentProjects\\chu_charlene_171105_thecowproblem.nb"];
```

Solution by Leonard Afeke (with help from DBN):

```
NotebookOpen[Directory[] <> "\\StudentProjects\\Afeke_Leonard_171102_CowProblem.nb"];
```

The 8-5-3 Milk Jug Problem

Challenge. Draw the state graph of this problem.

Gaussian Elimination - The Commutative Case

```
vs = Table[(i + j)^5, {i, 12}, {j, 8}];
vs // Column
```

```
{32, 243, 1024, 3125, 7776, 16 807, 32 768, 59 049}
{243, 1024, 3125, 7776, 16 807, 32 768, 59 049, 100 000}
{1024, 3125, 7776, 16 807, 32 768, 59 049, 100 000, 161 051}
{3125, 7776, 16 807, 32 768, 59 049, 100 000, 161 051, 248 832}
{7776, 16 807, 32 768, 59 049, 100 000, 161 051, 248 832, 371 293}
{16 807, 32 768, 59 049, 100 000, 161 051, 248 832, 371 293, 537 824}
{32 768, 59 049, 100 000, 161 051, 248 832, 371 293, 537 824, 759 375}
{59 049, 100 000, 161 051, 248 832, 371 293, 537 824, 759 375, 1 048 576}
{100 000, 161 051, 248 832, 371 293, 537 824, 759 375, 1 048 576, 1 419 857}
{161 051, 248 832, 371 293, 537 824, 759 375, 1 048 576, 1 419 857, 1 889 568}
{248 832, 371 293, 537 824, 759 375, 1 048 576, 1 419 857, 1 889 568, 2 476 099}
{371 293, 537 824, 759 375, 1 048 576, 1 419 857, 1 889 568, 2 476 099, 3 200 000}
```

```
MatrixRank[vs]
```

```
6
```

The Commutative Analog. Let $V = \mathrm{span}(v_1, \ldots, v_\alpha)$ be a subspace of $\mathbb{R}^n$. Before you die, understand V.

Solution: Gaussian Elimination. Prepare an empty table,

1	2	3	4	$\cdots$	$n-1$	n

Space for a vector $u_4 \in V$, of the form $u_4 = (0,0,0,1,*,\ldots,*)$; 1 :="the pivot".

Feed $v_1, \ldots, v_\alpha$ in order. To feed a non-zero v, find its pivotal position i.

1. If box i is empty, put v there.
2. If box i is occupied, find a combination v' of v and u_i that eliminates the pivot, and feed v'.

```
n = 8
```

```
8
```

```
Do[u[i] = Null, {i, n}]
```

```
Feed[v_List] /; v == Table[0, {n}] := Null;
Feed[v_List] /; Total[Abs[v]] == 0 := Null;
Feed[{0 ...}] := Null;
```

```
Feed[v_List] := Module[{i},
  i = 1; While[v[[i]] == 0, ++i];
  If[u[i] === Null,
   u[i] = v / v[[i]],
   Feed[v - v[[i]] u[i]]
  ]
 ]
```

```
Feed /@ vs
```

$$\left\{\left\{1, \frac{243}{32}, 32, \frac{3125}{32}, 243, \frac{16807}{32}, 1024, \frac{59049}{32}\right\},\right.$$
$$\left\{0, 1, \frac{148832}{26281}, \frac{510543}{26281}, \frac{1351744}{26281}, \frac{3035525}{26281}, \frac{6073056}{26281}, \frac{11148907}{26281}\right\},$$
$$\left\{0, 0, 1, \frac{3135671}{590048}, \frac{633941}{36878}, \frac{12796175}{295024}, \frac{3454435}{36878}, \frac{107106853}{590048}\right\},$$
$$\left\{0, 0, 0, 1, \frac{2486096}{460379}, \frac{88763170}{5064169}, \frac{223656560}{5064169}, \frac{481470395}{5064169}\right\}, \left\{0, 0, 0, 0, 1, \frac{5873}{1038}, \frac{3278}{173}, \frac{16891}{346}\right\},$$
$$\left.\{0, 0, 0, 0, 0, 1, 6, 21\}, \text{Null}, \text{Null}, \text{Null}, \text{Null}, \text{Null}, \text{Null}\right\}$$

```
Table[u[i], {i, n}]
```

$$\left\{\left\{1, \frac{243}{32}, 32, \frac{3125}{32}, 243, \frac{16807}{32}, 1024, \frac{59049}{32}\right\},\right.$$
$$\left\{0, 1, \frac{148832}{26281}, \frac{510543}{26281}, \frac{1351744}{26281}, \frac{3035525}{26281}, \frac{6073056}{26281}, \frac{11148907}{26281}\right\},$$
$$\left\{0, 0, 1, \frac{3135671}{590048}, \frac{633941}{36878}, \frac{12796175}{295024}, \frac{3454435}{36878}, \frac{107106853}{590048}\right\},$$
$$\left\{0, 0, 0, 1, \frac{2486096}{460379}, \frac{88763170}{5064169}, \frac{223656560}{5064169}, \frac{481470395}{5064169}\right\},$$
$$\left.\left\{0, 0, 0, 0, 1, \frac{5873}{1038}, \frac{3278}{173}, \frac{16891}{346}\right\}, \{0, 0, 0, 0, 0, 1, 6, 21\}, \text{Null}, \text{Null}\right\}$$

```
r = 0; Do[If[u[i] =!= Null, ++r], {i, n}]; r
```

6

```
Solve[x^2 == 9, {x}]
```

{{x → -3}, {x → 3}}

```
x^2 == 9
```

$x^2 == 9$

```
x^2 === 9
```

False

```
? ...
```

p... or RepeatedNull[*p*] is a pattern object
that represents a sequence of zero or more expressions, each matching *p*.
RepeatedNull[*p*, *max*] represents from 0 to *max* expressions matching *p*.
RepeatedNull[*p*, {*min*, *max*}] represents between *min* and *max* expressions matching *p*. ≫

```
? MatchQ
```

MatchQ[*expr*, *form*] returns True if the pattern *form* matches *expr*, and returns False otherwise.
MatchQ[*form*] represents an operator form of MatchQ that can be applied to an expression. ≫

```
MatchQ[7, _Integer]
```

True

```
MatchQ[7.5, _Integer]
```

False

```
MatchQ[{0, 0, 0, 1}, {_Integer ...}]
```

True

```
MatchQ[{0, 0, 0, 1}, {0 ...}]
```

False

```
MatchQ[{0, 0, 0, 0}, {0 ...}]
```

True

```
? Position
```

Position[*expr*, *pattern*] gives a list of the positions at which objects matching *pattern* appear in *expr*.
Position[*expr*, *pattern*, *levelspec*] finds only objects that appear on levels specified by *levelspec*.
Position[*expr*, *pattern*, *levelspec*, *n*] gives the positions of the first *n* objects found.
Position[*pattern*] represents an operator form of Position that can be applied to an expression. ≫

```
? FirstPosition
```

FirstPosition[*expr*, *pattern*] gives the position of the first element in *expr* that matches *pattern*, or Missing["NotFound"] if no such element is found.
FirstPosition[*expr*, *pattern*, *default*] gives *default* if no element matching *pattern* is found.
FirstPosition[*expr*, *pattern*, *default*, *levelspec*] finds only objects that appear on levels specified by *levelspec*.
FirstPosition[*pattern*] represents an operator form of FirstPosition that can be applied to an expression. ≫

```
rob = {1, 2, 2, 3, 4, 4, 4, 2, 6};
rob //. {lft___, x_, x_, rgt___} :> {lft, x, rgt}
```

```
{1, 2, 3, 4, 2, 6}
```

```
Union@rob
```

```
{1, 2, 3, 4, 6}
```

```
?DeleteDuplicates
```

```
DeleteDuplicates[list] deletes all duplicates from list.
DeleteDuplicates[list, test] applies test to pairs
   of elements to determine whether they should be considered duplicates.  >>
```

On to Rubik's Cube

```
?Cycles
```

```
Cycles[{cyc1, cyc2, ...}] represents a permutation with disjoint cycles cyci. >>
```

```
n = 54;
g1 = Cycles[{{1, 18, 45, 28}, {2, 27, 44, 19}, {3, 36, 43, 10}, {46, 52, 54, 48}, {47, 49, 53, 51}}];
g2 = Cycles[{{7, 16, 39, 30}, {8, 25, 38, 21}, {9, 34, 37, 12}, {13, 15, 33, 31}, {14, 24, 32, 22}}];
g3 = Cycles[{{28, 31, 34, 48}, {29, 32, 35, 47}, {30, 33, 36, 46}, {37, 39, 45, 43}, {38, 42, 44, 40}}];
g4 = Cycles[{{1, 3, 9, 7}, {2, 6, 8, 4}, {10, 54, 16, 13}, {11, 53, 17, 14}, {12, 52, 18, 15}}];
g5 = Cycles[{{1, 13, 37, 46}, {4, 22, 40, 49}, {7, 31, 43, 52}, {10, 12, 30, 28}, {11, 21, 29, 19}}];
g6 = Cycles[{{3, 48, 39, 15}, {6, 51, 42, 24}, {9, 54, 45, 33}, {16, 18, 36, 34}, {17, 27, 35, 25}}];
```

Non-Commutative Gaussian Elimination

Prepare a mostly-empty table,



Feed $g_1, \dots, g_\alpha$ in order. To feed a non-identity σ, find its pivotal position i and let $j := \sigma(i)$.

1. If box (i, j) is empty, put σ there.
2. If box (i, j) contains $\sigma_{i,j}$, feed $\sigma' := \sigma_{i,j}^{-1}\sigma$.

The Twist. When done, for every occupied (i, j) and (k, l), feed $\sigma_{i,j}\sigma_{k,l}$. Repeat until the table stops changing.

```
?PermutationProduct
```

```
PermutationProduct[a, b, c] gives the product of permutations a, b, c.  >>
```

```
PermutationProduct[g1, g2]
```

```
Cycles[{{1, 18, 45, 28}, {2, 27, 44, 19}, {3, 36, 43, 10}, {7, 16, 39, 30}, {8, 25, 38, 21},
  {9, 34, 37, 12}, {13, 15, 33, 31}, {14, 24, 32, 22}, {46, 52, 54, 48}, {47, 49, 53, 51}}]
```

```
PermutationProduct[g1, g3]
```

```
Cycles[{{1, 18, 43, 10, 3, 46, 52, 54, 28}, {2, 27, 40, 38, 42, 44, 19},
  {29, 32, 35, 47, 49, 53, 51}, {30, 33, 36, 37, 39, 45, 31, 34, 48}}]
```

```
a_ ∘ b_ := PermutationProduct[a, b]
```

```
? InversePermutation
```

InversePermutation[*perm*] returns the inverse of permutation *perm*. ≫

```
? PermutationSupport
```

PermutationSupport[*perm*] returns the support of the permutation *perm*. ≫

```
g1
```

```
Cycles[{{1, 18, 45, 28}, {2, 27, 44, 19}, {3, 36, 43, 10}, {46, 52, 54, 48}, {47, 49, 53, 51}}]
```

```
PermutationSupport[g1]
```

```
{1, 2, 3, 10, 18, 19, 27, 28, 36, 43, 44, 45, 46, 47, 48, 49, 51, 52, 53, 54}
```

```
Union @@ Union @@ g1
```

```
{1, 2, 3, 10, 18, 19, 27, 28, 36, 43, 44, 45, 46, 47, 48, 49, 51, 52, 53, 54}
```

```
? PermutationReplace
```

PermutationReplace[*expr*, *perm*] replaces each part in *expr* by its image under the permutation *perm*.
PermutationReplace[*expr*, *gr*] returns the list of images of *expr* under all elements of the permutation group *gr*. ≫

```
g1
```

```
Cycles[{{1, 18, 45, 28}, {2, 27, 44, 19}, {3, 36, 43, 10}, {46, 52, 54, 48}, {47, 49, 53, 51}}]
```

```
PermutationReplace[19, g1]
```

```
2
```

```
Remove[σ];
σ_ ∘ τ_ := PermutationProduct[τ, σ];
Feed[Cycles[{}]] := 1 + 1;
Feed[τ_] := Module[{i, j, k, l},
   i = Min[PermutationSupport[τ]];
   j = PermutationReplace[i, τ];
   If[Head[σ_{i,j}] === Cycles,
    Feed[InversePermutation[σ_{i,j}] ∘ τ],
    (*Else*) σ_{i,j} = τ; Pause[0.05];
    For[k = 1, k < n, ++k,
     For[l = k + 1, l ≤ n, ++l,
      If[Head[σ_{k,l}] === Cycles,
       Feed[σ_{i,j} ∘ σ_{k,l}]; Feed[σ_{k,l} ∘ σ_{i,j}]]
     ]]
   ]];
$RecursionLimit = 10^6;
```

```
ArrayPlot[
  Table[
   Which[
    Head[σ_{i,j}] === Cycles ∨ i == j, Red, i < j, Yellow, i > j, White],
   {j, n}, {i, n}]],
  Mesh → True, Frame → True, FrameTicks → Range@n, AspectRatio → 9/16, ImageSize → 512
 ] // Dynamic
```

```
Table[Feed[g_i]; Product[
  1 + Length[Select[Range[i + 1, n], Head[σ_{i,#}] === Cycles &]],
  {i, 1, n - 1}
 ], {i, 6}]
```

{4, 16, 159 993 501 696 000, 21 119 142 223 872 000, 43 252 003 274 489 856 000, 43 252 003 274 489 856 000}

Challenge. Move the bottom face of a Rubik's cube by only moving the other 5 faces.

```
? Select
```

Select[*list*, *crit*] picks out all elements e_i of *list* for which *crit*[e_i] is True.
Select[*list*, *crit*, *n*] picks out the first *n* elements for which *crit*[e_i] is True.
Select[*crit*] represents an operator form of Select that can be applied to an expression. ≫

```
? Range
```

Range[i_{max}] generates the list {1, 2, …, i_{max}}.
Range[i_{min}, i_{max}] generates the list {i_{min}, …, i_{max}}.
Range[i_{min}, i_{max}, *di*] uses step *di*. ≫

```
Product[
 1 + Length[Select[Range[i + 1, n], Head[σ_{i,#}] === Cycles &]],
 {i, 1, n - 1}
]
```

43 252 003 274 489 856 000

```
g_1
```

Cycles[{{1, 18, 45, 28}, {2, 27, 44, 19}, {3, 36, 43, 10}, {46, 52, 54, 48}, {47, 49, 53, 51}}]

```
Feed[g_1]
```

```
ArrayPlot[
 Table[
  Which[
   Head[σ_{i,j}] === Cycles ∨ i == j, Red, i < j, Yellow, i > j, White],
  {j, n}, {i, n}],
 Mesh → True, Frame → True, FrameTicks → Range@n, AspectRatio → 9/16, ImageSize → 512
]
```

```
Feed[g_2]
```

```
ArrayPlot[
 Table[
  Which[
   Head[σ_{i,j}] === Cycles ∨ i == j, Red, i < j, Yellow, i > j, White],
  {j, n}, {i, n}],
 Mesh → True, Frame → True, FrameTicks → Range@n, AspectRatio → 9/16, ImageSize → 512
]
```



```
g_2
```

```
Cycles[{{7, 16, 39, 30}, {8, 25, 38, 21}, {9, 34, 37, 12}, {13, 15, 33, 31}, {14, 24, 32, 22}}]
```

```
Feed[g_3]
```

```
ArrayPlot[
 Table[
  Which[
   Head[σ_{i,j}] === Cycles ∨ i == j, Red, i < j, Yellow, i > j, White],
  {j, n}, {i, n}],
 Mesh → True, Frame → True, FrameTicks → Range@n, AspectRatio → 9/16, ImageSize → 512
]
```